In[2]:=
```
SetDirectory["~/KappaLib/"];
<< kappaLib-1.2.m
<< helper.m
```

```
Loading KappaLib v1.2

Loading helper.m..
```

# In this notebook we define a medium kappa such that:

# 1) kappa is algebraically decomposable

# 2) beta^2-alpha gamma = 0

# 3) the Fresnel polynomial always has a linear factor, but it does not factorise into a product of two second order polynomials

- **Define medium**

In[5]:=
$$\text{kappa} = \text{emMatrixToKappa}\left[\begin{pmatrix} -3 & 1 & 1 & 1 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 1 & 0 \\ 1 & 0 & 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}\right];$$

In[6]:=
```
(* kappa is invertible *)
FullSimplify[emDet[kappa]]

(* kappa has no axion component *)
Simplify[emTrace[kappa]]

(* Since this does not simplify to {0}, kappa has a skewon component *)
Union[Flatten[kappa - emPoincare[kappa]]]

(* Since this does not simplify to {0}, kappa has a principal component *)
Union[Flatten[kappa + emPoincare[kappa]]]
```

Out[6]= 1

Out[7]= 0

Out[8]= {-4, -1, 0, 1, 4}

Out[9]= {-2, -1, 0, 1, 2}

### ■ Define constants alpha, beta, gamma and bivectors A and B:

In[10]:=
```
alpha = 1;
beta = -1;
gamma = 1;
rho = 1 / 2;
```

$$\text{Abivector} = \begin{pmatrix} 0 & 1 & 0 & 0 \\ -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix};$$

$$\text{Bbivector} = \begin{pmatrix} 0 & -4 & 1 & 1 \\ 4 & 0 & 0 & 0 \\ -1 & 0 & 0 & 1 \\ -1 & 0 & -1 & 0 \end{pmatrix};$$

### ■ Note these satisfy beta^2-alpha gamma = 0

In[16]:=
```
beta ^ 2 - alpha gamma
```

Out[16]= 0

### ■ Verify that kappa is algebraically decomposable:

In[17]:=
```
LHS = alpha emIdentityKappa[] +
    beta (kappa + emPoincare[kappa]) + gamma emCompose[emPoincare[kappa], kappa];
RHS = emBiProduct[rho, Abivector, Bbivector] + emBiProduct[rho, Bbivector, Abivector];
Union[Simplify[Flatten[(LHS - RHS)]]]
```

Out[19]= {0}

### ■ Compute Fresnel polynomial

In[20]:=
```
coords = {xi0, xi1, xi2, xi3};
fresnel = FullSimplify[emKappaToFresnel[kappa, coords]]
```

Out[21]=
$$\mathrm{xi2}\left(\mathrm{xi0}\left(\mathrm{xi1}\,(\mathrm{xi0}+\mathrm{xi1})+2\,(\mathrm{xi0}+3\,\mathrm{xi1})\,\mathrm{xi2}+\mathrm{xi2}^2\right)-\left(\mathrm{xi1}^2+\mathrm{xi1}\,\mathrm{xi2}+\mathrm{xi2}\,(3\,\mathrm{xi0}+\mathrm{xi2})\right)\mathrm{xi3}+\mathrm{xi2}\,\mathrm{xi3}^2\right)$$

### ■ Fresnel polynomial is product of linear factor and 3rd order factor

In[22]:=
```
factor1 = xi2;
factor2 = (xi0 (xi1 (xi0 + xi1) + 2 (xi0 + 3 xi1) xi2 + xi2^2) -
    (xi1^2 + xi1 xi2 + xi2 (3 xi0 + xi2)) xi3 + xi2 xi3^2);
Simplify[fresnel - factor1 factor2]
```

Out[24]= 0

### ■ Show that the Fresnel polynomial does not factor into a product of two second order polynomials

In[25]:=
```
gPlus = emMatrix["g", 4, Structure → "Symmetric"];
gMinus = emMatrix["h", 4, Structure → "Symmetric"];
gPlus // MatrixForm
gMinus // MatrixForm
factorised = (coords.gPlus.coords) (coords.gMinus.coords);
```

Out[27]//MatrixForm=

$$\begin{pmatrix} g11 & g12 & g13 & g14 \\ g12 & g22 & g23 & g24 \\ g13 & g23 & g33 & g34 \\ g14 & g24 & g34 & g44 \end{pmatrix}$$

Out[28]//MatrixForm=

$$\begin{pmatrix} h11 & h12 & h13 & h14 \\ h12 & h22 & h23 & h24 \\ h13 & h23 & h33 & h34 \\ h14 & h24 & h34 & h44 \end{pmatrix}$$

In[30]:=
```
constraints = Union[Flatten[CoefficientList[factorised - fresnel, coords]]];
constraints = simp[constraints];
show[constraints]
```

Out[32]//MatrixForm=

| | | |
|---|---|---|
| 1 | : | 0 |
| 2 | : | g11 h11 |
| 3 | : | g22 h22 |
| 4 | : | g33 h33 |
| 5 | : | g44 h44 |
| 6 | : | 2 (g12 h11 + g11 h12) |
| 7 | : | 2 (g13 h11 + g11 h13) |
| 8 | : | 2 (g14 h11 + g11 h14) |
| 9 | : | 2 (g22 h12 + g12 h22) |
| 10 | : | 2 (g23 h22 + g22 h23) |
| 11 | : | 2 (g24 h22 + g22 h24) |
| 12 | : | 2 (g33 h23 + g23 h33) |
| 13 | : | 2 (g44 h14 + g14 h44) |
| 14 | : | 2 (g44 h24 + g24 h44) |
| 15 | : | 2 (g44 h34 + g34 h44) |
| 16 | : | 1 + 2 g34 h33 + 2 g33 h34 |
| 17 | : | - 1 + 2 g33 h13 + 2 g13 h33 |
| 18 | : | g22 h11 + 4 g12 h12 + g11 h22 |
| 19 | : | g33 h22 + 4 g23 h23 + g22 h33 |
| 20 | : | g44 h11 + 4 g14 h14 + g11 h44 |
| 21 | : | g44 h22 + 4 g24 h24 + g22 h44 |
| 22 | : | - 2 + g33 h11 + 4 g13 h13 + g11 h33 |
| 23 | : | - 1 + g44 h33 + 4 g34 h34 + g33 h44 |
| 24 | : | 2 (g24 h11 + 2 g14 h12 + 2 g12 h14 + g11 h24) |
| 25 | : | 2 (2 g24 h12 + g22 h14 + g14 h22 + 2 g12 h24) |
| 26 | : | 2 (g34 h11 + 2 g14 h13 + 2 g13 h14 + g11 h34) |
| 27 | : | 2 (g44 h12 + 2 g24 h14 + 2 g14 h24 + g12 h44) |
| 28 | : | 2 (g44 h13 + 2 g34 h14 + 2 g14 h34 + g13 h44) |
| 29 | : | 2 (g44 h23 + 2 g34 h24 + 2 g24 h34 + g23 h44) |
| 30 | : | 3 + 4 g34 h13 + 2 g33 h14 + 2 g14 h33 + 4 g13 h34 |
| 31 | : | 1 + 2 g34 h22 + 4 g24 h23 + 4 g23 h24 + 2 g22 h34 |
| 32 | : | 1 + 4 g34 h23 + 2 g33 h24 + 2 g24 h33 + 4 g23 h34 |
| 33 | : | - 1 + 2 g23 h11 + 4 g13 h12 + 4 g12 h13 + 2 g11 h23 |
| 34 | : | - 1 + 4 g23 h12 + 2 g22 h13 + 2 g13 h22 + 4 g12 h23 |
| 35 | : | 2 (- 3 + g33 h12 + 2 g23 h13 + 2 g13 h23 + g12 h33) |
| 36 | : | 4 (g34 h12 + g24 h13 + g23 h14 + g14 h23 + g13 h24 + g12 h34) |

- **If the Fresnel polynomial factorises the above equations must have a solution g_ij and h_ij.**
- **By D.Cox, J.Little, D.O'Shea "Ideals, Varieties, and Algorithms" we know that a system of polynomial equations do not have have a solution (in the complex domain) if a Gröbner basis for the equations is {1}. This is the case here:**

In[33]:=
```
gb = GroebnerBasis[constraints, Variables[constraints]]; // Timing
```

Out[33]= {0.920789, Null}

In[34]:=
```
gb
```

Out[34]= {1}

- **Thus equations ‘constraints’ have no solution for h_ij and g_ij and there is no factorisation for the Fresnel surface of kappa into quadratic forms.**

---

# Show that D=A satisfies the equation

## D (kappa + beta Id) = 1/2 trace (rho bar(D) otimes D) A + B.

In[35]:=
```
(*
 * If D=1/2 D^ij d/dx^i /\ d/dx^j is a bivector we represent
 * the coefficients by the anti-symmetric matrix with coefficients
 * of (D^ij)_ij. If kappa is an antisymmetric (2,2)-tensor,
 * then this routine returns coefficients of bivector D (kappa).
 *)
contract[biv_,kappa_] := Table[
 1/2 Sum[biv[[i]][[j]]emReadNormal[kappa,a,b,i,j]
       ,
       {i, 1, 4},{j, 1, 4}
     ]
 ,
 {a, 1, 4}, {b, 1, 4}
]
```

In[36]:=
```
Dbivector = Abivector;
dLHS = contract[Dbivector, kappa + beta emIdentityKappa[]];
dRHS = 1 / 2 emTrace[emBiProduct[rho, Dbivector, Dbivector]] Abivector + Bbivector;
deqs = Union[Flatten[Simplify[dLHS - dRHS]]]
```

Out[39]= {0}